

Unbound Subreports

This sample demonstrates how to set up unbound subreports by passing information and search criteria from the parent report to the Subreport. This sample demonstrates two different methods for passing the Recordset. One method is to clone the recordset and the second is to establish a new Recordset in the Subreport. When working with unbound reports, it is important to use the DataInitialize and FetchData events to add fields to the report's field collection and retrieve data from the database.

1. Add three new ActiveReports to a new project as shown in the first tutorial and set their properties as follows:

Name	rptCustomers	rptCustomersClone	rptOrders
-------------	--------------	-------------------	-----------

2. Select the most current version of Microsoft's ActiveX Data Object Library in Visual Basic's reference list.

With rptCustomers

1. Remove the PageHeader/Footer section.
2. Insert a GroupHeader/Footer section.
3. Set the GroupHeader's properties as follows:

Name	ghCountry
DataField	Country
Height	810
CanShrink	True
GrpKeepTogether	2-All
KeepTogether	True
Repeat	1-Every Page
Font.Bold	True

4. Place a textbox in the ghCountry section and set its properties as follows:

Name	txtCountry
DataField	Country
Height	540
Left	0
Top	0
Width	2880
Font.Size	22
Font.Bold	True

5. Place three labels in the ghCountry section and set their properties as follows:

Name	lblCustomer	lblCity	lblPostalCode
Caption	Customer	City	Postal Code
Height	270	270	270
Left	0	2880	5760
Top	540	540	540
Width	2880	2800	1800
Font.Bold	True	True	True

6. Set the GroupFooter's height to 60.
7. Set the Detail section's properties as follows:

Height	630
KeepTogether	True

8. Place a Subreport control in the Detail section and set its properties as follows:

Name	subCustomersClone
Height	630
Left	0
Top	0
Width	9360

9. Add the following code to rptCustomers:

```
Public cn As ADODB.Connection
Dim rs As ADODB.Recordset
Public myCloneRS As ADODB.Recordset
Private s_Country As String

Private Sub ActiveReport_DataInitialize()
    Dim cnnStr As String
    ' Open data base here (called from form)
    Set cn = New ADODB.Connection
    Set rs = New ADODB.Recordset
    Set myCloneRS = New ADODB.Recordset

    ' Add fields to ActiveReports' Fields Collection
    Fields.Add "Country"
    ' Set Connection String and connect to DB
    cnnStr = "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=C:\Program _
        Files\Microsoft Visual Studio\VB98\NWind.mdb"
    cn.Open cnnStr

    ' Set recordset properties and generate recordset
    rs.CursorType = adOpenStatic
    rs.LockType = adLockOptimistic
    rs.Open "Select * from customers order by Country", cn

    ' Create a clone of the current recordset to use in the subreport
    Set myCloneRS = rs.Clone
End Sub

Private Sub ActiveReport_FetchData(EOF As Boolean)
    ' Exit sub if EOF is true
    If rs.EOF Then Exit Sub

    ' Sets ActiveReports' fields collection values to the
    ' current recordset value
    Fields("Country").Value = rs!country
    s_Country = rs!country
    ' Advance recordset
    rs.MoveNext

    If Not rs.EOF Then
        ' If not EOF set the FetchData's EOF value to False
        EOF = False
    End If
End Sub

Private Sub ActiveReport_ReportEnd()
    ' Unload the subreport
    Unload subCustomersClone.object
    Set subCustomersClone.object = Nothing
    ' Unload the recordsets and database
    Set myCloneRS = Nothing
    Set rs = Nothing
    Set cn = Nothing
End Sub

Private Sub ActiveReport_ReportStart()
    ' Set the subreport's object to a new rptCustomersClone
    Set subCustomersClone.object = New rptCustomersClone
End Sub

Private Sub Detail_Format()
    ' Filter the cloned recordset to select the country country
    myCloneRS.Filter = "Country = '" & s_Country & "'"
End Sub
```

With rptCustomersClone

- 1. Remove the PageHeader/Footer section.
- 2. Set the Detail section's properties as follows:

Height	1260
BackClor	&H00FFC0C0&
BackStyle	1-Normal
CanShrink	True
KeepTogether	True

- 3. Place three textboxes in the Detail section and set their properties as follows:

Name	txtCustomer	txtCity	txtPostalCode
DataField	Customer	City	PostalCode
Height	270	270	270
Left	0	2880	5760
Top	0	0	0
Width	2880	2880	1800

4. Place four labels in the Detail section and set their properties as follows:

Name	lblOrdered	lblRequired	lblShipped	lblCost
Caption	Ordered	Required	Shipped	Cost
Height	270	270	270	270
Left	90	1530	2880	4320
Top	360	360	360	360
Width	1350	1260	1350	1440
Font.Bold	True	True	True	True

5. Place a Subreport control in the Detail section and set its properties as follows:

Name	subOrders
Height	630
Left	0
Top	630
Width	9360

6. Add the following code to rptCustomers:

```
Private rsClone As ADODB.Recordset
Private s_CustomerID As String

Private Sub ActiveReport_DataInitialize()
    Fields.RemoveAll
    ' Add fields to ActiveReports' Fields Collection
    Fields.Add "Customer"
    Fields.Add "City"
    Fields.Add "Country"
    Fields.Add "PostalCode"
    Set rsClone = New ADODB.Recordset
    Set rsClone = rptCustomer.myCloneRS

    ' Move to the beginning of the cloned recordset
    If Not rsClone.EOF Or Not rsClone.BOF Then
        rsClone.MoveFirst
    End If
End Sub

Private Sub ActiveReport_FetchData EOF As Boolean)
    ' Exit sub if EOF is true
    If rsClone.EOF Then Exit Sub

    ' Sets ActiveReports' fields collection values to the
    ' current recordset value
    Fields("Customer").Value = rsClone!CompanyName
    Fields("City").Value = rsClone!City
    Fields("Country").Value = rsClone!country
    Fields("PostalCode").Value = rsClone!PostalCode
    s_CustomerID = rsClone!CustomerID
    ' Advance recordset
    rsClone.MoveNext

    If Not rsClone.EOF Then
        ' If not EOF set the FetchData's EOF value to False
        EOF = False
    End If
End Sub

Private Sub ActiveReport_ReportStart()
    ' set the subreport's object to a new rptOrders
    Set subOrders.object = New rptOrders
End Sub

Private Sub Detail_Format()
    ' Pass the current customer ID to the rptOrders subreport
    subOrders.object.CustomerID = s_CustomerID
End Sub
```

With rptOrders

1. Remove the PageHeader/Footer section.
2. Insert a GroupHeader/Footer section.
3. Set the GroupHeader's properties as follows:

Name	ghOrders
Height	60
BackColor	&H000040C0&
BackStyle	1-Normal
GrpKeepTogether	2-All

KeepTogether	True
---------------------	------

4. Set the GroupFooter's properties as follows:

Name	gfOrders
Height	60
BackColor	&H000040C0&
BackStyle	1-Normal
KeepTogether	True

5. Set the Detail section's properties as follows:

Height	270
BackColor	&H00C0E0FF&
BackStyle	1-Normal
CanShrink	True
KeepTogether	True

6. Place four textboxes in the Detail section and set their properties as follows:

Name	txtOrdered	txtRequired	txtShipped	txtCost
Text	Ordered	Required	Shipped	Cost
DataField	Ordered	Required	Shipped	Cost
Height	270	270	270	270
Left	0	1440	2880	4320
Top	0	0	0	0
Width	1440	1440	1440	1440
OutputFormat				Currency

7. Add the following code to rptOrders:

```
Private rsNewRS As ADODB.Recordset
Private m_CID As String

Public Property Let CustomerID(s_CID As String)
    ' Store the current customerID
    m_CID = s_CID
End Property

Public Property Get CustomerID() As String
    ' Get the current customerID
    CustomerID = m_CID
End Property

Private Sub ActiveReport_DataInitialize()
    Fields.RemoveAll
    ' Add fields to ActiveReports' Fields Collection
    Fields.Add "Ordered"
    Fields.Add "Shipped"
    Fields.Add "Required"
    Fields.Add "Cost"
    'Generate a new orders recordset based off of the current customerID
    Set rsNewRS = New ADODB.Recordset
    rsNewRS.CursorType = adOpenForwardOnly
    rsNewRS.LockType = adLockOptimistic
    rsNewRS.Open "SELECT * FROM Orders WHERE CustomerID = '" & CustomerID_
        & "'", rptCustomer.cn
End Sub

Private Sub ActiveReport_FetchData.EOF As Boolean)
    ' Exit sub if EOF is true
    If rsNewRS.EOF Then Exit Sub

    ' Sets ActiveReports' fields collection values to the
    ' current recordset value
    Fields("Ordered").Value = rsNewRS!OrderDate
    Fields("Shipped").Value = rsNewRS!ShippedDate
    Fields("Required").Value = rsNewRS!RequiredDate
    Fields("Cost").Value = rsNewRS!Freight

    ' Advance recordset
    rsNewRS.MoveNext
    If Not rsNewRS.EOF Then
        ' If not EOF set the FetchData's EOF value to False
        EOF = False
    End If
End Sub
```

8. Run the report.